

[8/26/00; updated for Mathcad Professional 2000; and for Mathcad 2001 Professional]

Mathcad Primer: Mathcad is a friendly and powerful way of doing mathematics, particularly practical mathematics relevant to physics, chemistry, and engineering. Here we give some basic instructions and examples.

First, all this text is contained in a text region, which is selected by the option "Text Region" under the symbol "Insert" in the menu at the top of the document. Text regions allow you to write comments in simple English anywhere on the page. You can't do mathematics in a text region, and you can't write text outside a text region (where you do mathematics).

Now, to business...

(1) Equality: most times when you would write an = sign in a mathematical equation, you use another symbol ":= " in mathcad. This is produced by typing the colon character (outside a text region!). Basically, this symbol means "is defined as". so, for example to define the function  $y(x)=2\cdot x + 1$ , you type:

$$y(x) := 2 \cdot x + 1$$

The only time you type an = sign in Mathcad is when you want to examine the value of a variable ("register") or evaluate a function at a particular value of the independent variable. Once you've defined the function you can evaluate it at a particular value of its argument.

so:

$$p := 2 \cdot 3 + 60 \quad \text{Defines } p$$

$$p = 66 \quad \text{Evaluates } p$$

$$y(1) = 3 \quad \text{Evaluates } y, \text{ defined above, at value } x=1$$

Range Variables: The way you execute a loop (let input values vary over a specified range) in mathcad is with "range variables". To loop over a range of real numbers, you type:

$$x := 1, 1.2..3 \quad \text{This will let } x \text{ run over the values } 1, 1.2, 1.4...2.8, 3 \text{ in whatever expression(s) come after this statement.}$$


$y(x) =$	$x =$
3	1
3.4	1.2
3.8	1.4
4.2	1.6

Typing the = key, e.g. " $y(x) =$ " generates the values of the function at the  $x$  values specified in the loop above.

4.6	1.8
5	2
5.4	2.2
5.8	2.4
6.2	2.6
6.6	2.8
7	3

An important point about the range variable statement for real valued variables: the ".." symbol is achieved by typing the semicolon key.

Graphing functions: under "Insert" (see menu at top) is the choice "Graph", followed by "X-Y plot". This generates a graph which can be used with the range variable statement to make standard x-y plots.




Function  $y(x)$  defined above plotted over range defined above.

It is easy to put more than one function plot on the same graph. For example, define:

$$yy(x) := x^2$$

Note: to get a superscript like "square", type "^".

Type a comma after entering "y(x)". this gives you a second blank slot, into which you type "y2(x)"



Some examples and practice:


(1) Exercise: plot the functions  $y(x)=x^2$  and  $y(x)=x^4$  on the interval  $[-1:1]$ . (Experiment with the increment between x values to get "smooth" curves for your plots.)

(2) Mathcad has lots of "special functions" built in. For example, trigonometric functions like sin and cos. Here is how to plot  $\sin(x)$  over one period:

$f(x) := \sin(x)$

$x := 0, .05 .. 2 \cdot \pi$

Note: To get greek letters type the roman letter equivalent, e.g. p, then type "cntrl G". greek letters can be used to name variables and functions. Mathcad knows what  $\pi$  is. Also, note that if your increment in x does not quite "work" [i.e.  $2 \cdot \pi$  is not a multiple of .05], Mathcad can still cope.



Excercise: Plot  $\cos(x)$  between 0 and  $2\pi$ ; plot  $\sin(x)$  over the same interval on the same graph.

Last Exercise: If a ball is thrown upward with an initial velocity 10 m/sec, it's vertical position  $y$  in meters (with respect to the point of release) as a function of time is given by the formula:  $y(t)=10*t-4.9*t^2$ . Plot the vertical position of the ball as a function of time over the first 2.5 seconds of its trajectory.

More Business: Subscripts, integer range variables...

It is often useful to have a variable subscripted by an integer (to denote components). To get a subscript placeholder type the character "[". This generates expressions like:

$$x_2 := 1$$

You can loop over an integer range variable like this:

$$j := 0..5$$

N.B.: The symbol ".." in the live region is obtained by typing a semicolon (";")

$$x_j := 2 \cdot j$$


$$y_j := x_j + 4$$

$$y_j =$$

4
6
8
10
12
14

$$x_j =$$

0
2
4
6
8
10



Matrix Manipulations: A matrix is just a square array of numbers.

To construct a matrix in Mathcad, click on View, Toolbars, then Matrix. Then click on the matrix place-holder icon. You will then be asked to select the number of rows and columns in the matrix.

In this way, you can construct the following 3x3 matrix:

$$A := \begin{pmatrix} -1 & 1 & 0 \\ 1 & -2 & 1 \\ 0 & 1 & -3 \end{pmatrix}$$

A few comments about matrix specification: You can address a particular element by two subscripts:

$$A_{2,2} = -3$$

Note that the indices begin at 0, not 1 as in usual mathematical convention, in the default operation of Mathcad. (You can reset the ORIGIN=1 if you wish.)

Also note that you can display an entire column of a matrix using the Cntrl-^ keys. You can extract a particular element of this using the subscript key ("["):

$$A^{\langle 0 \rangle} = \begin{pmatrix} -1 \\ 1 \\ 0 \end{pmatrix} \qquad (A^{\langle 0 \rangle})_2 = 0$$

A vector is just a matrix with one column. For example:

$$v := \begin{pmatrix} 1 \\ 2 \\ 4 \end{pmatrix}$$

Multiplying matrices is easy. Just use the multiplication symbol ("\*"). [Don't forget, the number of columns in the 1st matrix must equal the number of rows in the second.]

$$A \cdot v = \begin{pmatrix} 1 \\ 1 \\ -10 \end{pmatrix}$$

Now for some more interesting stuff... Mathcad inverts matrices "automatically". Just raise the matrix to the -1 power:

$$A^{-1} = \begin{pmatrix} -2.5 & -1.5 & -0.5 \\ -1.5 & -1.5 & -0.5 \\ -0.5 & -0.5 & -0.5 \end{pmatrix}$$

Check that Mathcad has done the inversion correctly. Does  $A^{-1} \cdot A = 1$ ?:

$$A \cdot A^{-1} = \begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

With this feature, it is trivial to solve any system of linear equations, i.e. sets of N equations of the form  $A \cdot x = v$ , where A is an NxN matrix, v is an N dimensional vector, and  $x = (x_1, x_2, \dots, x_N)^t$  is a (column) vector containing N unknowns. Then:

$$x := A^{-1} \cdot v \quad x = \begin{pmatrix} -7.5 \\ -6.5 \\ -3.5 \end{pmatrix}$$

Another important type of computation that Mathcad performs automatically is the determination of eigenvalues and eigenvectors of a matrix. These quantities enter into many types of fundamental theory (the Schrodinger Eq., normal modes of vibration, etc.). For any square matrix A, the internal routine "eigenvals" returns a vector containing all eigenvalues. The routine "eigenvecs" returns all the corresponding eigenvectors.

$$w := \text{eigenvals}(A) \quad w = \begin{pmatrix} -0.268 \\ -2 \\ -3.732 \end{pmatrix}$$

$$B := \text{eigenvecs}(A) \quad B = \begin{pmatrix} 0.789 & 0.577 & 0.211 \\ 0.577 & -0.577 & -0.577 \\ 0.211 & -0.577 & 0.789 \end{pmatrix}$$

Thus, the first column of B corresponds to the first entry in w, etc.

Note that the routine  $\text{eigenvec}(A, w_1)$  can be used to find one eigenvector (the one corresponding to eigenvalue  $w_1$ , which must be found by calling eigenvals).

$$v_1 := \text{eigenvec}(A, -0.268) \quad v_1 = \begin{pmatrix} 0.789 \\ 0.577 \\ 0.211 \end{pmatrix}$$

$$v_2 := \text{eigenvec}(A, -2) \quad v_2 = \begin{pmatrix} -0.577 \\ 0.577 \\ 0.577 \end{pmatrix}$$

Note that the Mathcad routine gives unit normalized eigenvectors.  
[Aside: for real symmetric matrices, all eigenvectors are orthogonal.]

$$v_1 \cdot v_1 = 1 \quad v_1 \cdot v_2 = 0$$

Note that the eigenvalues and eigenvectors of a matrix A can be used to decompose it into the following product:

$$B \cdot \begin{pmatrix} w_0 & 0 & 0 \\ 0 & w_1 & 0 \\ 0 & 0 & w_2 \end{pmatrix} \cdot B^{-1} = \begin{pmatrix} -1 & 1 & 0 \\ 1 & -2 & 1 \\ 0 & 1 & -3 \end{pmatrix}$$

Basic operations involving random numbers and statistics:

The workhorse of Monte Carlo methods is the uniform random number generator.

The built-in Mathcad function  $\text{rnd}(x)$  generates a uniform random number between 0 and x:

$$\text{rnd}(1) = 1.268 \times 10^{-3}$$

Or, to generate an array of N uniform random numbers in the range (a,b) use  $\text{runif}(N,a,b)$ :

$$\text{runif}(5, -1, 1) = \begin{pmatrix} -0.613 \\ 0.17 \\ -0.299 \\ 0.646 \\ -0.652 \end{pmatrix}$$

To get a feeling for how a uniform random generator, it is instructive to bin the output obtained from  $\text{runif}$  into a histogram:



$N := 2000$

$v := \text{runif}(N + 1, 0, 1)$

$x_{\min} := 0 \quad x_{\max} := 1 \quad n_{\text{int}} := 40 \quad dx := \frac{x_{\max} - x_{\min}}{n_{\text{int}}}$


$j := 0..n_{\text{int}}$

$\text{int}_j := x_{\min} + j \cdot dx$

$f := \text{hist}(\text{int}, v)$

Note: the mathcad function hist creates a histogram of the data contained in array v.

$j := 0..n_{\text{int}} - 1$



[Note the bar-graph plotting mode here: select "bar" as the Type of trace]

It is often useful to examine the mean and standard deviation of a distribution of outcomes. Internal Mathcad functions make it easy to do this:

$$j := 0..nint - 1$$

$$f1_j := \frac{f_j}{N \cdot dx}$$

$$\text{mean}(f1) = 1$$

$$\text{stdev}(f1) = 0.139$$


Finding a real root of a function of one variable using the Mathcad routine "root":

$$a := 1 \quad b := -6 \quad c := 11 \quad d := -6$$

$$f(x) := a \cdot x^3 + b \cdot x^2 + c \cdot x + d$$

$$x := .5, .55..3.5$$

Consider, as an example, the cubic equation indicated. The real-valued roots of this equation are the intersections with the line  $f(x)=0$ .



$$x := 1.9$$

The function "root" requires an initial guess. It then locks into the root which is closest to this guess..

$$x0 := \text{root}(f(x), x) \quad x0 = 2$$

The Mathcad Programming Palette:

$$\text{quad}(a, b, c) := \begin{array}{l} \text{rt} \leftarrow \sqrt{b^2 - 4 \cdot a \cdot c} \\ v_0 \leftarrow \frac{-b + \text{rt}}{2 \cdot a} \\ v_1 \leftarrow \frac{-b - \text{rt}}{2 \cdot a} \\ v \end{array}$$

A simple example: finding the roots of the quadratic equation

$$a \cdot x^2 + b \cdot x + c = 0$$

Note: the output returned is the variable, array, or matrix obtained on the final line of the "subroutine".

$$\text{quad}(1, 2, 10) = \begin{pmatrix} -1 + 3i \\ -1 - 3i \end{pmatrix}$$

Here's a slightly more complicated program: find the roots of a general cubic equation:

$$a := 1 \quad b := -6 \quad c := 11 \quad d := -6$$

$$f(x) := a \cdot x^3 + b \cdot x^2 + c \cdot x + d$$

Record the cubic function  $f(x)$  with specified parameters.

```

cubrt :=
  x ← 4
  x0 ← root(f(x), x)
  A ← a
  B ← b + a·x0
  C ← c + b·x0 + a·x02
  rt ← √(B2 - 4·A·C)
  v1 ← (-B + rt) / (2·A)
  v2 ← (-B - rt) / (2·A)
  v0 ← x0
  v

```

$$\text{cubrt} = \begin{pmatrix} 3 \\ 2 \\ 1 \end{pmatrix}$$

x := 0, .02.. 3.5

Find one root of f(x) numerically using the Mathcad function "root". [There is always one real-valued root to a cubic equation. It doesn't matter which of the three roots is located, so the seed value used with root is arbitrary.]

Factor out the root x0 by "longhand division". This leaves the quadratic equation:

$$A \cdot x^2 + B \cdot x + C = 0$$

Solve this using the quadratic formula, thus determining the remaining 2 roots of the f(x), whether they are both real or form a complex-conjugate pair.

For visual confirmation, plot:

